
1. DAX Fundamentals
Data Analysis Expressions (DAX) is the formula language for Power BI semantic models. Understanding DAX concepts is essential for creating effective measures and calculations.
1.1 Key Concepts
	Concept
	Description

	Filter Context
	Current filters applied from slicers, visuals, and relationships

	Row Context
	Current row being evaluated in a table iteration

	Context Transition
	Converting row context to filter context (via CALCULATE)

	Evaluation Order
	Filters apply, then measure calculates



1.2 Measure vs Calculated Column
	Aspect
	Measure
	Calculated Column

	When Calculated
	Query time
	Refresh time

	Storage
	Formula only
	Stored in model

	Context
	Filter context
	Row context

	Best For
	Aggregations
	Categorization



Warning: Prefer measures over calculated columns for aggregations to reduce model size.


2. Essential Patterns
2.1 CALCULATE Pattern
CALCULATE modifies filter context for a calculation.
// Add filter
Active Member Claims = 
CALCULATE(
    [Total Claims],
    dim_member[status] = "Active"
)

// Remove filter
All Claims = 
CALCULATE(
    [Total Claims],
    ALL(dim_date)
)
2.2 DIVIDE Pattern
Safe division with zero handling.
// Always use DIVIDE for ratios
Denial Rate = 
DIVIDE(
    [Denied Claims],
    [Total Claims],
    0  -- Return 0 if denominator is 0
)

// Bad: Can cause divide by zero
// [Denied Claims] / [Total Claims]
2.3 Variables Pattern
Use variables for readability and performance.
// Good: Variables calculated once
Growth Rate = 
VAR CurrentPeriod = [Total Claims]
VAR PreviousPeriod = [PY Claims]
VAR Difference = CurrentPeriod - PreviousPeriod
RETURN
DIVIDE(Difference, PreviousPeriod, 0)

// Bad: Repeated calculation
// DIVIDE([Total Claims] - [PY Claims], [PY Claims], 0)
// (calculates [PY Claims] twice)


3. Time Intelligence
3.1 Basic Time Functions
// Year-to-date
YTD Claims = TOTALYTD([Total Claims], dim_date[date])

// Month-to-date
MTD Claims = TOTALMTD([Total Claims], dim_date[date])

// Quarter-to-date
QTD Claims = TOTALQTD([Total Claims], dim_date[date])
3.2 Period Comparison
// Previous year
PY Claims = 
CALCULATE([Total Claims], SAMEPERIODLASTYEAR(dim_date[date]))

// Previous month
PM Claims = 
CALCULATE([Total Claims], DATEADD(dim_date[date], -1, MONTH))

// Previous period (dynamic)
Prior Period = 
CALCULATE([Total Claims], PREVIOUSMONTH(dim_date[date]))
3.3 Year-over-Year
YoY Change = 
VAR Current = [Total Claims]
VAR Prior = [PY Claims]
RETURN
Current - Prior

YoY Change % = 
VAR Current = [Total Claims]
VAR Prior = [PY Claims]
RETURN
DIVIDE(Current - Prior, Prior, BLANK())
3.4 Rolling Periods
// Rolling 12 months
Rolling 12M Claims = 
CALCULATE(
    [Total Claims],
    DATESINPERIOD(dim_date[date], MAX(dim_date[date]), -12, MONTH)
)

// Rolling average
Rolling 3M Average = 
AVERAGEX(DATESINPERIOD(dim_date[date], MAX(dim_date[date]), -3, MONTH), [Total Claims])


4. Filter Functions
4.1 ALL Family
// Remove all filters from table
% of Total = 
DIVIDE([Total Claims], CALCULATE([Total Claims], ALL(fact_claims)))

// Remove filter from specific column
ALLEXCEPT Claims = 
CALCULATE([Total Claims], ALLEXCEPT(dim_member, dim_member[state]))

// All selected (respects slicers)
% of Selected = 
DIVIDE([Total Claims], CALCULATE([Total Claims], ALLSELECTED()))
4.2 FILTER Function
// Filter to specific condition
High Value Claims = 
CALCULATE(
    [Claim Count],
    FILTER(fact_claims, fact_claims[amount] > 10000)
)

// Better: Use column filter when possible
High Value Claims = 
CALCULATE(
    [Claim Count],
    fact_claims[amount] > 10000
)
4.3 KEEPFILTERS
// Add filter without overriding existing
Active High Value = 
CALCULATE(
    [Total Claims],
    KEEPFILTERS(fact_claims[amount] > 10000),
    dim_member[status] = "Active"
)


5. Iterator Functions
5.1 SUMX Pattern
// Row-by-row calculation
Total Extended Amount = 
SUMX(
    fact_claims,
    fact_claims[quantity] * fact_claims[unit_price]
)

// With filter
Active Extended = 
SUMX(
    FILTER(fact_claims, RELATED(dim_member[status]) = "Active"),
    fact_claims[quantity] * fact_claims[unit_price]
)
5.2 AVERAGEX and COUNTX
// Average of calculated values
Avg Claim per Member = 
AVERAGEX(
    VALUES(dim_member[member_id]),
    [Total Claims]
)

// Count with condition
High Value Count = 
COUNTX(
    FILTER(fact_claims, fact_claims[amount] > 10000),
    fact_claims[claim_id]
)
5.3 RANKX
// Rank members by claims
Member Rank = 
RANKX(
    ALL(dim_member),
    [Total Claims],
    ,
    DESC,
    DENSE
)


6. Performance Optimization
6.1 Best Practices
1. Use variables to avoid repeated calculations
1. Prefer simple column filters over FILTER()
1. Use KEEPFILTERS instead of FILTER when possible
1. Avoid iterators on large tables
1. Use DISTINCTCOUNT over COUNTROWS(VALUES())
1. Minimize context transitions
6.2 Anti-Patterns to Avoid
// Bad: Nested CALCULATE
// CALCULATE(CALCULATE([Measure], Filter1), Filter2)

// Good: Single CALCULATE with multiple filters
CALCULATE([Measure], Filter1, Filter2)

// Bad: FILTER on entire table
// CALCULATE([Total], FILTER(fact_claims, [column] = value))

// Good: Direct column filter
CALCULATE([Total], fact_claims[column] = value)
6.3 Debugging
1. Break complex measures into smaller parts
1. Use variables to inspect intermediate values
1. Test measures with specific filter context
1. Use DAX Studio for query analysis
1. Check Performance Analyzer in Power BI Desktop

7. Quick Reference
7.1 Common Functions
	Function
	Purpose

	SUM, AVERAGE
	Basic aggregations

	CALCULATE
	Modify filter context

	DIVIDE
	Safe division

	ALL, ALLEXCEPT
	Remove filters

	FILTER
	Row-level filtering

	SUMX, AVERAGEX
	Row-by-row calculation

	VALUES, DISTINCT
	Unique values

	RELATED
	Follow relationship

